

CHAPTER 4

FORMULA TECHNIQUES

InFocus

Most people are familiar with simple **formulas** such as =B2+B3 and simple **functions** such as SUM, but there is so much more that you can do with formulas and functions. Most functions require **parameters** – extra bits of information – to perform their task. You can actually use other functions or formulas as the parameters for a function – effectively nesting functions within functions.

In this session you will:

- ✓ gain an understanding of how to scope a formula
- ✓ gain an understanding of how to develop a nested function
- ✓ learn how to create a nested function
- ✓ learn how to edit a nested function
- ✓ learn how to copy nested functions
- ✓ learn how to use concatenation to join text and values
- ✓ learn how to switch to manual recalculation
- ✓ learn how to force a recalculation
- ✓ learn how to paste values from formulas.

SCOPING A FORMULA

A **nested function** is a function within a function, and they can be quite tricky to read and build unless you plan ahead. **Scoping** refers to the process of defining the problem or calculation so

that you can identify the different parts of the formula you need to create. The order in which these parts need to be calculated determines where each part is nested in the formula.

Scoping & Building Steps

Formulas that use nested functions are best built in several stages and should be thoroughly tested at each stage to ensure that they achieve the correct result. The steps are:

1. determine what the overall formula will calculate
2. break down the formula into its component parts
3. create the base function and ensure that it works
4. add the additional functions and elements of the formula one by one, testing each one as it is added to the overall formula.

Nesting Functions Workshop Example

The objective of the workshop example is to create a formula that calculates a dividend payable to ten superannuation investors. The spreadsheet includes the following information:

	A	B	C	D	E	F	G	H
1	The Alpheius Super Fund							
2				Investment	Risk (A)	Growth (B)	Cons'tive (C)	
3	Bonus Date	30/06/1998		\$1,000	3%	2%	1%	
4				\$5,000	4%	3%	1%	
5				\$10,000	5%	4%	2%	
6				\$15,000	6%	5%	2%	
7				\$20,000	7%	6%	4%	
8				\$25,000	8%	8%	5%	
9								
10								
11	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend	
12	Pedro	Kavana	Mr	15/05/1999	\$10,000	C		
13	Jessica	Dunn	Miss	12/04/1998	\$12,000	A		
14	Tim	Nyguen	Mr	13/05/1998	\$3,000	A		
15	Fabian	Considine	Mr	12/01/1998	\$12,500	C		

The dividend is calculated by applying a **percentage** to the original investment. The **percentage** is taken from a sliding scale and is determined by:

- the amount of the **investment**
- the investment **scale** that the investor chose
- the **date** of initial investment: as an incentive, a 5% bonus was offered to investors who signed up on or before June 30, 1998.

The base function in the formula will be a **VLOOKUP** function, which is written as follows:

VLOOKUP(lookup value, lookup table, return value)

The **lookup value** is the amount of the investment. The **lookup table** is the dividend table. The **return value** is the percentage that will be applied to the investment amount to calculate the dividend.

The return value, however, causes a problem as it depends upon the scale chosen by the investor. This problem is overcome by using an **IF** function to ascertain which scale was chosen and to use this information to select the appropriate value from the table.

Another **IF** function can be used to test whether the investor signed up within the bonus period and to apply the appropriate bonus amount.

DEVELOPING A NESTED FUNCTION

The best way to **develop a complex formula** is by developing each of the components first and then combining them. By writing each of the parts in sentence form, you will be able to understand

the logic of each more easily. You can then establish where the individual parts go in the overall scheme of your formula, create a base function, and then build your formula from there

Developing the Workshop Example

The overall formula for the workshop example can be stated as:

*=Investment Amount * (Dividend Percentage + Bonus)*

The Dividend Percentage

The logic of the **Dividend Percentage** calculation can be stated as:

Look up the investment amount in the dividend table

If the investor chose scale A, then return the percentage from column 2 in the table

If the investor chose scale B, then return the percentage from column 3

Otherwise, return the percentage from column 4

The base function used to calculate the **Dividend Percentage** can be written as:

VLOOKUP(investment amount, dividend table, return column)

The Investment Scale

We then need to add IF functions within the VLOOKUP to allow for the three scales. If there were only two scales, the IF function could be written as:

IF(scale chosen = A, return column 2, otherwise return column 3)

Because there are three scales, we need to add the second IF function to provide the extra choice. The final IF functions can be written as:

IF(scale chosen = A, return column 2, IF(scale chosen = B, return column 3, return column 4))

The Bonus

Finally, we need to determine whether or not to pay the **Bonus**. The logic of the **Bonus** calculation can be stated as:

If the joining date is earlier than the bonus date, then add 5%, otherwise add nothing

This can also be done using an IF function, written as:

IF(the joining date is earlier than the bonus date, pay 5%, otherwise don't pay anything)

Now all we need to do is translate these into Excel terminology and add cell references. We will start by creating the base function, so that it can be tested, and then we will add the additional parts of the formula.

CREATING NESTED FUNCTIONS

A **nested function is created** by placing a function as a parameter within another function. When you create nested functions the placement of brackets is especially important and should be

checked carefully. Excel helps you with this task by using different colours for matched pairs of brackets. You can also improve the readability of the formula by placing the parts on separate lines.

Try This Yourself:

Open File

Before starting this exercise you **MUST** open the file E818 Formula Techniques_1.xlsx...

- 1 Click in cell **G12**
- 2 Type
=VLOOKUP(E12,\$D\$3:\$G\$8,
- 3 Press **Alt** + **Enter** to create a new line – splitting the parts of the formula up makes it easier to read
- 4 Type **IF(F12="A",2,**
- 5 Press **Alt** + **Enter** to create a new line
- 6 Type **IF(F12="B",3,**
- 7 Type **4)))**
- 8 Press **Enter** to complete the formula
- 9 Click in cell **F12**, type **A** then press **Enter** to see the dividend rate for scale **A**

The final part of the formula is the 'false' part of the last IF statement...

The value 0.02 should appear – the rate taken from G5 which is the \$10,000 investment at Scale C...

3

Scale	Dividend
C	=VLOOKUP(E12,\$D\$3:\$G\$8,
A	
A	VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])
C	
C	

4

Joined	Investment	Scale	Dividend
/1999	\$10,000	A	=VLOOKUP(E12,\$D\$3:\$G\$8,
/1998	\$12,000	A	IF(F12="A",2,
/1998	\$3,000	A	IF(logical_test, [value_if_true], [value_if_false])
/1998	\$12,500	C	
/1998	\$4,500	C	

7

Joined	Investment	Scale	Dividend
/1999	\$10,000	A	=VLOOKUP(E12,\$D\$3:\$G\$8,
/1998	\$12,000	A	IF(F12="A",2,
/1998	\$3,000	A	IF(F12="B",3,4)))
/1998	\$12,500	C	
/1998	\$4,500	C	

8

Joined	Investment	Scale	Dividend
15/05/1999	\$10,000	C	0.02
12/04/1998	\$12,000	A	
13/05/1998	\$3,000	A	
12/01/1998	\$12,500	C	
13/02/1998	\$4,500	C	

9

Joined	Investment	Scale	Dividend
15/05/1999	\$10,000	A	0.05
12/04/1998	\$12,000	A	
13/05/1998	\$3,000	A	
12/01/1998	\$12,500	C	
13/02/1998	\$4,500	C	

For Your Reference...

To **create** a **nested function**:

1. Type the function as required
2. Use **Alt** + **Enter** to create new lines
3. Check the placement of brackets carefully

Handy to Know...

- Excel uses colour in formulas to indicate **cell references**. When you edit an existing formula, the cells referred to will appear outlined in a specific colour, and the corresponding cell references in the formula will appear in matching colours.

EDITING NESTED FUNCTIONS

Once the base function has been created and tested, the additional parts of the nested function can be added by **editing the formula**. Nested functions are edited the same way as any other

Excel formula. When you click on the cell, the formula is displayed in the formula bar and can then be modified. Excel also uses coloured references to help you understand the formula.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file E818 Formula Techniques_2.xlsx...

1

Click on **G12** to select it

2

Click immediately after the first equal sign in the **Formula bar**

This places the formula in edit mode and displays matched coloured references. For example, **\$D\$3:\$G\$8** appears in green in the formula and is outlined in green on the worksheet...

3

Type **E12***

This ensures that the investment amount will be multiplied by the Dividend percentage...

4

Click on at the right end of the **Formula bar** to expand it, then click at the end of the formula and press **Alt + Enter**

5

On the new line, type **+IF(D12<=\$B\$3,5%,0))**

This adds the bonus percentage if the start date is earlier than the bonus date...

6

Press **Enter** to complete the formula then click on to restore the formula bar to one line

2

	B	C	D	E	F	G	H
	ieius Super Fund						
			Investment	Risk (A)	Growth (B)	Cons'tive (C)	
30/06/1998			\$1,000	3%	2%	1%	
			\$5,000	4%	3%	1%	
			\$10,000	5%	4%	2%	
			\$15,000	6%	5%	2%	
			\$20,000	7%	6%	4%	
			\$25,000	8%	8%	5%	

	B	C	D	E	F	G	H
	Lastname	Title	Joined	Investment	Scale	Dividend	
	Kavana	Mr	15/05/1999	\$10,000	A	=VLOOKUP(E	
	Dunn	Miss	12/04/1998	\$12,000	A		

4

	B	C	D	E	F	G	H
	ieius Super Fund						
			Investment	Risk (A)	Growth (B)	Cons'tive (C)	

5

	B	C	D	E	F	G	H
	ieius Super Fund						
			Investment	Risk (A)	Growth (B)	Cons'tive (C)	

6

	B	C	D	E	F	G	H
	Lastname	Title	Joined	Investment	Scale	Dividend	
	Kavana	Mr	15/05/1999	\$10,000	A	\$500	
	Dunn	Miss	12/04/1998	\$12,000	A		
	Nyguen	Mr	13/05/1998	\$3,000	A		
	Considine	Mr	12/01/1998	\$12,500	C		
	Jovanovski	Ms	13/02/1998	\$4,500	C		

For Your Reference...

To **edit** a **nested function**:

- Click on the cell to select it
- Click in the formula in the **Formula bar** and edit as required
- Press **Enter** to complete the changes

Handy to Know...

- When you have finished editing a formula, you can click on the **Enter** button to enter the formula. If you make a mistake, you can click on the **Cancel** button or press **Esc** to exit from **Edit** mode.

COPYING NESTED FUNCTIONS


You can **copy formulas with nested functions** the same way that you copy any formula. However, you should be especially careful of the cell addresses. Relative cell addresses will adjust

automatically when you copy files, while absolute addresses will remain unchanged. These can be difficult to identify in a complex formula, so it is wise to check the results thoroughly after copying.


Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file *E818 Formula Techniques_3.xlsx...*


- 1 Click on **G12** then click on **Expand Formula Bar**  and examine the formula

The only absolute references are *\$D\$3:\$G\$8*, and the Bonus date *\$B\$3*. The rest are relative which will adjust as you copy the formula. These include the date joined, investment amount and scale...

- 2 Click on **Collapse Formula Bar**  to return it to one line

- 3 Move the mouse pointer to the fill handle at the bottom right corner of **G12** then double-click to copy the formula down the column

This technique for copying works as long as you have a continuous list on the left of the original cell...

- 4 Click on **G22** then double-click on the **AutoSum** tool  to calculate the total of the dividends

AutoSum will create a Sum function for the cells immediately above it

G12		fx		=E12*(VLOOKUP(E12,\$D\$3:\$G\$8, IF(F12="A",2, IF(F12="B",3,4))) +IF(D12<=\$B\$3,5%,0))				
	A	B	C	D	E	F	G	
1	The Alpheius Super Fund							
2				Investment	Risk (A)	Growth (B)	Cons'tive (C)	
3	Bonus Date	30/06/1998		\$1,000	3%	2%	1%	
4				\$5,000	4%	3%	1%	
5				\$10,000	5%	4%	2%	
6				\$15,000	6%	5%	2%	
7				\$20,000	7%	6%	4%	
8				\$25,000	8%	8%	5%	
9								
10								
11	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend	
12	Pedro	Kavana	Mr	15/05/1999	\$10,000	A	\$500	
13	Jessica	Dunn	Miss	12/04/1998	\$12,000	A		
14	Tim	Nyguen	Mr	13/05/1998	\$3,000	A		

1

3

Joined	Investment	Scale	Dividend
15/05/1999	\$10,000	A	\$500
12/04/1998	\$12,000	A	\$1,200
13/05/1998	\$3,000	A	\$240
12/01/1998	\$12,500	C	\$875
13/02/1998	\$4,500	C	\$270
14/03/1999	\$2,300	B	\$46
12/03/1999	\$1,200	A	\$36
1/03/1998	\$2,300	B	\$161
23/02/1998	\$15,000	C	\$1,050
21/01/1999	\$23,000	C	\$920

4

14/03/1999	\$2,300	B	\$46
12/03/1999	\$1,200	A	\$36
1/03/1998	\$2,300	B	\$161
23/02/1998	\$15,000	C	\$1,050
21/01/1999	\$23,000	C	\$920
			\$5,298

For Your Reference...

To **copy a nested function**:

1. Check the cell references
2. Copy the formula using your preferred copying technique

Handy to Know...

- When you double-click on a formula, it will place it in **Edit** mode. Cells referred to in the formula will appear with coloured outlines. If you want the formula to use exactly the same cells as those outlined, no matter where the formula is copied to, the cell references must be **absolute**.

USING CONCATENATION





Concatenation is the process of joining text and/or values together. This can be used to form a phrase or expression from existing data in the worksheet. The main character used for

concatenation is the ampersand **&**. All text must be enclosed in quotation marks. The **TEXT()** function can be used to convert dates and values into text so that they can be included in the result too.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file *E818 Formula Techniques_4.xlsx...*

- 1 Click on **H11** and type **Report**, then press 
- 2 In cell **H12**, type the following, using  +  to create new lines, and including the spaces before and after the words
=C12 & " " & B12 & " joined the fund on " &
- 3 Complete the rest of the function as shown
- 4 Press  to create the formula
- 5 Copy the formula to the range **H13:H21** and then click elsewhere, to deselect it

A report has been created for every investor

2

Scale	Dividend	Report
A	\$500	=C12 & " " & B12 &
A	\$1,200	" joined the fund on " &
A	\$240	
C	\$875	
C	\$270	
B	\$46	
A	\$36	
B	\$161	
C	\$1,050	
C	\$920	
	\$5,298	

3

Scale	Dividend	Report
A	\$500	=C12 & " " & B12 &
A	\$1,200	" joined the fund on " &
A	\$240	TEXT(D12,"dd/mm/yyyy") &
C	\$875	" with an investment of " &
C	\$270	TEXT(E12,"\$#,##0") &
B	\$46	". The dividend is now " &
A	\$36	TEXT(G12,"\$#,##0") & "."
B	\$161	
C	\$1,050	
C	\$920	
	\$5,298	

5

Dividend	Report
\$500	Mr Kavana joined the fund on 15/05/1999 with an investment of \$10,000. The dividend is now \$500.
\$1,200	Miss Dunn joined the fund on 12/04/1998 with an investment of \$12,000. The dividend is now \$1,200.
\$240	Mr Nyguen joined the fund on 13/05/1998 with an investment of \$3,000. The dividend is now \$240.
\$875	Mr Considine joined the fund on 12/01/1998 with an investment of \$12,500. The dividend is now \$875.
\$270	Ms Jovanovski joined the fund on 13/02/1998 with an investment of \$4,500. The dividend is now \$270.
\$46	Ms Beaumont joined the fund on 14/03/1999 with an investment of \$2,300. The dividend is now \$46.
\$36	Mr Underwood joined the fund on 12/03/1999 with an investment of \$1,200. The dividend is now \$36.
\$161	Ms Schenk joined the fund on 01/03/1998 with an investment of \$2,300. The dividend is now \$161.
\$1,050	Miss Perera joined the fund on 23/02/1998 with an investment of \$15,000. The dividend is now \$1,050.
\$920	Ms Rasheed joined the fund on 21/01/1999 with an investment of \$23,000. The dividend is now \$920.
\$5,298	

For Your Reference...

To **concatenate text** and **values**:

1. Use ampersand (&) to add items
2. Enclose actual text in quotation marks, including spaces
3. Convert dates and values using **TEXT()**

Handy to Know...

- You could use the report created by the concatenation in a **mail merge** using Microsoft Word to produce individual report letters for each client.

SWITCHING TO MANUAL RECALCULATION


Recalculation refers to processing the formulas in a spreadsheet to calculate new results. Formulas are usually recalculated each time a value in a dependent cell changes, but you can


turn off **automatic** recalculation and instead set Excel to **manual**. This means that no formulas will be recalculated unless you specifically request Excel to perform the calculations.

Try This Yourself:

Same File


Continue using the previous file with this exercise, or open the file *E818 Formula Techniques_5.xlsx*...

1 Click on **F13** and type **B** then press 
The dividend and total will update to reflect the change...

2 Click on the **Formulas** tab in the **Ribbon** and click on the **Calculation Options** tool  to display the menu

You can see that it's set to **Automatic**...


3 Select **Manual**, then click in **F16**

4 Type **A** and press 

You'll notice that nothing changes, even though scale A gives a different percentage and should result in a higher dividend

1

	Joined	Investment	Scale	Dividend	Report
	15/05/1999	\$10,000	A	\$500	Mr Kavana joined the fur
	12/04/1998	\$12,000	B	\$1,080	Miss Dunn joined the fur
	13/05/1998	\$3,000	A	\$240	Mr Nyguen joined the fu
	12/01/1998	\$12,500	C	\$875	Mr Considine joined the
	13/02/1998	\$4,500	C	\$270	Ms Jovanovski joined the
	14/03/1999	\$2,300	B	\$46	Ms Beaumont joined the
	12/03/1999	\$1,200	A	\$36	Mr Underwood joined th
	1/03/1998	\$2,300	B	\$161	Ms Schenk joined the fur
	23/02/1998	\$15,000	C	\$1,050	Miss Perera joined the fu
	21/01/1999	\$23,000	C	\$920	Ms Rasheed joined the f
				\$5,178	




4

	Joined	Investment	Scale	Dividend	Report
	15/05/1999	\$10,000	A	\$500	Mr Kavana joined the fur
	12/04/1998	\$12,000	B	\$1,080	Miss Dunn joined the fur
	13/05/1998	\$3,000	A	\$240	Mr Nyguen joined the fu
	12/01/1998	\$12,500	C	\$875	Mr Considine joined the
	13/02/1998	\$4,500	A	\$270	Ms Jovanovski joined the
	14/03/1999	\$2,300	B	\$46	Ms Beaumont joined the
	12/03/1999	\$1,200	A	\$36	Mr Underwood joined th
	1/03/1998	\$2,300	B	\$161	Ms Schenk joined the fur
	23/02/1998	\$15,000	C	\$1,050	Miss Perera joined the fu
	21/01/1999	\$23,000	C	\$920	Ms Rasheed joined the f
				\$5,178	

For Your Reference...

To **switch to manual recalculation**:

1. Click on the **Formulas** tab
2. Click on **Calculation Options** 
3. Select **Manual**

Handy to Know...

- The **Calculation** settings are global and will affect every spreadsheet you work with. You can't force only one workbook to require manual recalculation without affecting others (unless you've done some fancy programming, but that's another story).

FORCING A RECALCULATION

If you turn off manual recalculation, you will need to **force recalculation** at some stage to update the formulas. There are several options to choose from for recalculation. You can

recalculate any changed formulas in the current worksheet or all open workbooks, recalculate all formulas in all workbooks irrespective of changes, or check all formulas before recalculating.

Try This Yourself:


Open File

Before starting this exercise you **MUST** open the file E818 Formula Techniques_6.xlsx...

1

Click on **F20** and examine the scale, dividend and total amounts

2

Type **A** and press 


Nothing happens...

3

On the **Formulas** tab, click on **Calculate Now** 

The formulas will be recalculated and the results will be updated...

4

Click on **Calculation Options**  and select **Automatic**

Now recalculation will happen automatically when values are changed

1

Joined	Investment	Scale	Dividend	Report
15/05/1999	\$10,000	A	\$500	Mr Kavana joined the fur
12/04/1998	\$12,000	B	\$1,080	Miss Dunn joined the fur
13/05/1998	\$3,000	A	\$240	Mr Nyguen joined the fu
12/01/1998	\$12,500	C	\$875	Mr Considine joined the
13/02/1998	\$4,500	A	\$360	Ms Jovanovski joined the
14/03/1999	\$2,300	B	\$46	Ms Beaumont joined the
12/03/1999	\$1,200	A	\$36	Mr Underwood joined th
1/03/1998	\$2,300	B	\$161	Ms Schenk joined the fur
23/02/1998	\$15,000	C	\$1,050	Miss Perera joined the fu
21/01/1999	\$23,000	C	\$920	Ms Rasheed joined the fe
			\$5,268	

2


Joined	Investment	Scale	Dividend	Report
15/05/1999	\$10,000	A	\$500	Mr Kavana joined the fur
12/04/1998	\$12,000	B	\$1,080	Miss Dunn joined the fur
13/05/1998	\$3,000	A	\$240	Mr Nyguen joined the fu
12/01/1998	\$12,500	C	\$875	Mr Considine joined the
13/02/1998	\$4,500	A	\$360	Ms Jovanovski joined the
14/03/1999	\$2,300	B	\$46	Ms Beaumont joined the
12/03/1999	\$1,200	A	\$36	Mr Underwood joined th
1/03/1998	\$2,300	B	\$161	Ms Schenk joined the fur
23/02/1998	\$15,000	A	\$1,050	Miss Perera joined the fu
21/01/1999	\$23,000	C	\$920	Ms Rasheed joined the fe
			\$5,268	

3


Joined	Investment	Scale	Dividend	Report
15/05/1999	\$10,000	A	\$500	Mr Kavana joined the fur
12/04/1998	\$12,000	B	\$1,080	Miss Dunn joined the fur
13/05/1998	\$3,000	A	\$240	Mr Nyguen joined the fu
12/01/1998	\$12,500	C	\$875	Mr Considine joined the
13/02/1998	\$4,500	A	\$360	Ms Jovanovski joined the
14/03/1999	\$2,300	B	\$46	Ms Beaumont joined the
12/03/1999	\$1,200	A	\$36	Mr Underwood joined th
1/03/1998	\$2,300	B	\$161	Ms Schenk joined the fur
23/02/1998	\$15,000	A	\$1,650	Miss Perera joined the fu
21/01/1999	\$23,000	C	\$920	Ms Rasheed joined the fe
			\$5,868	

For Your Reference...

To **force a recalculation**:

1. Click on the **Formulas** tab
2. Click on **Calculate Now**  or press **F9**

Handy to Know...

- To recalculate the current worksheet, click on **Calculate Sheet**  on the **Formulas** tab or press **Shift + F9**. To recalculate all workbooks irrespective of changes, press **Ctrl + Alt + F9**. To check the formulas as well as recalculate, press **Ctrl + Shift + Alt + F9**.

PASTING VALUES FROM FORMULAS

Sometimes it's useful to be able to grab the results of a calculation and use the value elsewhere in a spreadsheet without keeping the formula. For example, you may want to keep a

copy of some data at a certain point in time, knowing that it won't change at a later date. You can do this by copying formulas and **pasting only the values**.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file *E818 Formula Techniques_7.xlsx...*

1 Select the range **B12:B21** then hold down and select **G12:G21**

The first part of the range contains text and the second contains formulas...

2 Click on the **Home** tab on the **Ribbon** and click on **Copy**

3 Click on **B24**, then click on the drop arrow for **Paste** and click on **Values**

4 Press to hide the copy marquees, then click on **C24** and check the entry in the **Formula bar**

You'll see that only the value 500 has been pasted – the formula isn't retained

1

	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26
		Firstname	Lastname	Title	Joined	Investment	Scale	Dividend									
		Pedro	Kavana	Mr	15/05/1999	\$10,000	A	\$500	Mr								
		Jessica	Dunn	Miss	12/04/1998	\$12,000	B	\$1,080	Ms								
		Tim	Nyguen	Mr	13/05/1998	\$3,000	A	\$240	Mr								
		Fabian	Considine	Mr	12/01/1998	\$12,500	C	\$875	Mr								
		Rose	Jovanovski	Ms	13/02/1998	\$4,500	A	\$360	Ms								
		India	Beaumont	Ms	14/03/1999	\$2,300	B	\$46	Ms								
		Bryn	Underwood	Mr	12/03/1999	\$1,200	A	\$36	Mr								
		Sylvia	Schenk	Ms	1/03/1998	\$2,300	B	\$161	Ms								
		Courtney	Perera	Miss	23/02/1998	\$15,000	A	\$1,650	Ms								
		Shivanthe	Rasheed	Ms	21/01/1999	\$23,000	C	\$920	Ms								
								\$5,868									

3

	A	B	C	D	E	F	G	
		Firstname	Lastname	Title	Joined	Investment	Scale	Dividend
		Pedro	Kavana	Mr	15/05/1999	\$10,000	A	\$500
		Jessica	Dunn	Miss	12/04/1998	\$12,000	B	\$1,080
		Tim	Nyguen	Mr	13/05/1998	\$3,000	A	\$240
		Fabian	Considine	Mr	12/01/1998	\$12,500	C	\$875
		Rose	Jovanovski	Ms	13/02/1998	\$4,500	A	\$360
		India	Beaumont	Ms	14/03/1999	\$2,300	B	\$46
		Bryn	Underwood	Mr	12/03/1999	\$1,200	A	\$36
		Sylvia	Schenk	Ms	1/03/1998	\$2,300	B	\$161
		Courtney	Perera	Miss	23/02/1998	\$15,000	A	\$1,650
		Shivanthe	Rasheed	Ms	21/01/1999	\$23,000	C	\$920
								\$5,868
			Kavana	500				
			Dunn	1080				
			Nyguen	240				
			Considine	875				
			Jovanovski	360				
			Beaumont	46				
			Underwood	36				
			Schenk	161				
			Perera	1650				
			Rasheed	920				

4

	A	B	C	D	E	F	G	
		Sylvia	Schenk	Ms	1/03/1998	\$2,300	B	\$161
		Courtney	Perera	Miss	23/02/1998	\$15,000	A	\$1,650
		Shivanthe	Rasheed	Ms	21/01/1999	\$23,000	C	\$920
								\$5,868
			Kavana	500				
			Dunn	1080				

For Your Reference...

To **paste values** from **formulas**:

1. Click on the formula
2. Click on **Copy**
3. Click in the destination cell
4. Click on the drop arrow for **Paste** and click on **Values**

Handy to Know...

- When you copy formulas, you have the option to paste formulas, values and links. A link is a reference to the cell containing the formula. For example, if the cell containing the copied formula is G12, the link created by pasting will be =G\$12.